FORM version 4.2

B. Ruijla,b, T. Uedaa and J.A.M. Vermaserena

aNikhef Theory Group, Science Park 105, 1098 XG Amsterdam, The Netherlands
bLeiden Centre of Data Science, Leiden University, Niels Bohrweg 1, 2333 CA Leiden, The Netherlands

Abstract

We introduce FORM 4.2, a new minor release of the symbolic manipulation toolkit. We demonstrate several new features, such as a new pattern matching option, new output optimization, and automatic expansion of rational functions.
1 Introduction

We introduce a new version of the symbolic manipulation toolkit FORM \cite{1,2}, named FORM 4.2. FORM aims to be a high-performance symbol manipulator, even in cases where there are billions of terms that take up several terabytes of disk space.

This release is an update from FORM 4.1, which was released in 2013. In the latest release we have fixed more than 50 bugs and have introduced more than 20 new features. For an overview of all changes made from FORM 4.1 to FORM 4.2, please see the Changelog \cite{3}. The driving force of the development of new features and improvements in FORM is the use of the program in actual research projects. Many of the new features in FORM 4.2 were inspired by the use in FORCER \cite{4} but should be useful in other environments as well. The correctness, efficiency and limitations of the new features have been extensively tested through its physics applications.

The latest version of FORM can be obtained from GitHub \cite{5}, where installation instructions can also be found. The installation of FORM, including the threaded version TForm \cite{6}, on x86-64 machines with normal Linux distributions should be rather straightforward.

Whenever new features are introduced, users will resort to creative applications that are beyond the imagination of the designers. This can mean that certain restrictions, that were originally thought to never cause problems, can become an obstacle to such innovative use. Hence, if the user encounters issues, please report them to the GitHub Issue Tracker \cite{7}. One can also search for known issues in the Issue Tracker.

The layout of this work is as follows. In section \ref{section:new_features} we show some new features and in section \ref{section:summary} we give a summary.

2 Notable new features

In this section we present some notable new features of FORM 4.2. For more details of all features available in FORM 4.2, please see the reference manual \cite{8}.

2.1 Generating all matches

The Identify statement, often abbreviated as id, is one of core statements of FORM. This statement tries to match a given pattern and performs algebraic replacements. FORM version 4.2 introduces a new option all, which tries to generate all possible matches instead of just the first. For example,

\begin{verbatim}
CF v,f,s;
L F = v(1,2,3,4);
 id all v(?a,?b) = f(?a)*s(?b);
 Print +s;
 .end
\end{verbatim}

gives the following result:

\begin{verbatim}
F =
 + f*s(1,2,3,4)
 + f(1)*s(2,3,4)
\end{verbatim}
Figure 1: Topology for which we find all automorphisms with id all.

\[ + f(1,2)*s(3,4) \]
\[ + f(1,2,3)*s(4) \]
\[ + f(1,2,3,4)*s \]

The id all statement can be used in many different scenarios. One is to find all automorphisms (symmetries) of a graph, described by the vertex structure vt, which have (undirected) momenta assigned to all edges connected to the vertex as function arguments \[9\].

In figure 1 we show a graph for which we find all eight automorphisms using the following code:

```csharp
V Q,Q1,Q2,p1,...,p8;
CF map,vx(s);  * note that the vertex structure vx is symmetric
L F = vx(Q1,p1,p6)*vx(p1,p2,p7)*vx(p2,p3,p8)*
   vx(p3,p4,Q2)*vx(p4,p5,p7)*vx(p5,p6,p8);

id all vx(Q1?,p1?,p6?)*vx(p1?,p2?,p7?)*vx(p2?,p3?,p8?)*
   vx(p3?,p4?,Q2?)*vx(p4?,p5?,p7?)*vx(p5?,p6?,p8?) =
   map(Q1,Q2,p1,p2,p3,p4,p5,p6,p7,p8);

Print +s;
.end
```

which yields

\[ F = \]
\[ + \text{map}(Q1,Q2,p1,p2,p3,p4,p5,p6,p7,p8) \]
\[ + \text{map}(Q1,Q2,p1,p7,p4,p3,p8,p6,p2,p5) \]
\[ + \text{map}(Q1,Q2,p6,p5,p4,p3,p2,p1,p8,p7) \]
\[ + \text{map}(Q2,Q1,p3,p8,p6,p1,p7,p4,p2,p5) \]
\[ + \text{map}(Q2,Q1,p4,p5,p6,p1,p2,p3,p7,p8) \]
\[ + \text{map}(Q2,Q1,p4,p7,p1,p6,p8,p3,p5,p2) \]

The map function provides a mapping from the old edge labeling to a new one, which leaves the graph invariant.
2.2 Output optimization

Output optimization of polynomials was introduced in FORM 4.1 [10]. It relies on the idea that the number of arithmetic operations required to evaluate a polynomial can be reduced by pulling variables outside of brackets (Horner’s rule) and by removing common subexpressions. Finding the best order in which to extract variables (a Horner scheme) is difficult and FORM has different algorithms for it, namely Format O1 to Format O4.

In FORM 4.2 the performance of the output optimization has been increased, by improving the common subexpression detection. The mode that gives the best results in FORM 4.1, Format O3, has been improved to be less dependent on a user-given exploration-exploitation constant, based on the work of [11] and [12]. Furthermore, a new algorithm has been added, based on local search methods [13]. This option is called Format O4 and it generally produces better results and is faster than the Monte Carlo Tree Search method used in Format O3. An example is presented below:

```plaintext
S a,b,c,d,e,f,g,h,i,j,k,l,m,n;
L G = (4*a^4+b+c+d + i^4 + g*n^3)^10 +
    (a*h + e + f*i*j + g + h)^8 + (i + j + k + l + m + n)^12;
Format O4,saIter=300;  # use 300 iterations for optimization
.sort
#optimize G
#write "Optimized with Horner scheme: 'optimscheme_'"
#write "Number of operations in output: 'optimvalue_'"
#clearoptimize
.end
```

When running with tform, every thread runs its own search and the best value is selected. This gives better results than using fewer cores with a higher number of optimization iterations, saIter.

2.3 Automatic series expansion of rational functions

The Polyratfun, which was introduced in FORM 4.0 to treat multivariate rational functions as coefficients of terms, now supports automatic expansion in one variable. In the example below the rational function represented by the function rat is expanded with respect to the variable ep up to ep^5:

```plaintext
S ep;
CF rat;
Polyratfun rat;
L F = rat(ep + 1, ep^2 + 3*ep + 1);
.sort
Polyratfun rat(expand, ep, 5);
(Print;
.end
```

The code yields:
\[ F = \text{rat}(1 - 2\times ep + 5\times ep^2 - 13\times ep^3 + 34\times ep^4 - 89\times ep^5); \]

All manipulations performed in the expansion mode will truncate the series quickly. As a result, the expansion mode may be faster than the unexpanded mode if the coefficients become complicated. Additionally, it can avoid the intrinsic \text{MaxTermSize} restriction of \text{FORM}.

### 2.4 Textual manipulation on the output by dictionaries

Occasionally output from computer algebra systems provides input for another program, which may require suitable translation. In \text{FORM}, the \text{Format} statement can be used for preparing the output in a format of another program, for example, \text{Fortran}, \text{C} and so on. \text{FORM} version 4.2 gives another way of controlling the output to a certain extent: \text{dictionaries}. A dictionary is a collection of pairs of a \text{word} and its translation. The word can be a variable, a number, a function with specific arguments or a special character like the multiplication sign (*). The translation can be any string enclosed in double quotes. For example,

\begin{verbatim}
S x1,x2,x3;
L F = (x1+x2+x3)^2;
.sort
#opendictionary texdict
#add x1: "x_1"
#add x2: "x_2"
#add x3: "x_3"
#add *: " "
#closedictionary
#usedictionary texdict
#write "%E",F
.end
\end{verbatim}

prints the expression \( F \) for the \LaTeX{} math mode:

\[ x_3^2 + 2 \times x_2 \times x_3 + x_2^2 + 2 \times x_1 \times x_3 + 2 \times x_1 \times x_2 + x_1^2 \]

Here \#opendictionary texdict ... \#closedictionary defines a dictionary with the name texdict. Each \#add instruction defines a pair of a word and its translation. The \#usedictionary instruction sets the predefined dictionary for the textual manipulation on the output. Then, each word in the dictionary is replaced with its translation in the \#write instruction.

### 2.5 Spectators

Sometimes a large number of terms is passed through one or more modules for which the user knows that they will remain unmodified. Processing these irrelevant terms will cause overhead, including a potentially heavy sort cost.

The Spectator system can be used to remove terms from the current expression and to store them in a buffer, called a Spectator file. These terms will be ignored until they
are copied back into an active expression. In essence, the Spectator system works as a filter. Below is an example:

```plaintext
S x;
CF f,g;

CreateSpectator FSpec "Fspec.spec";
L F = f(1);
.sort

#do i=1,10000
  id f(x?) = theta_(10000-x)*f(x + 1) + g(x);  * term blow-up
  if (count(g,1)) ToSpectator FSpec;  * filter the g terms
  .sort:round'i';
#endo

CopySpectator F1 = FSpec;
.sort
RemoveSpectator FSpec;
.end
```

The above code takes about 0.25 seconds to run. If the line with ToSpectator is commented out, it takes about 13 seconds to run.

### 2.6 Zero-dimensional sparse tables as pure functions

High-level programming languages are mainly classified into two groups: imperative languages and declarative languages. Modern programming languages, however, tend to have aspects of both the groups and evolve towards multi-paradigm languages, which makes some programming tasks easier or more intuitive. FORM, which is clearly imperative, is not an exception. The users are now allowed to declare tables without any table indices. Such zero-dimensional (sparse) tables can have function arguments with wildcarding, which effectively leads to user-defined pure functions and opens a way of functional programming up to some extent. For example,

```plaintext
S n;
Table fac(n?pos0_);
Fill fac = delta_(n) + theta_(n) * n * fac(n-1);

L F = fac(5);
Print;
.end
```

implements a function fac as

\[
fac(n) = \begin{cases} 
1 & n = 0, \\
 n \times fac(n - 1) & n \geq 1.
\end{cases}
\] (1)

for a non-negative integer n, and yields
F = 120;

Of course in this case one can use the more efficient built-in fac_ function\footnote{In addition to the performance issue, the user-defined function suffers from a restriction with the current implementation of FORM: the recursion depth is limited by the stack size.}

A more complicated example is given below:

```FORM
S n,n1,n2;
Table fib(n?int_);
Table fibimpl(n?int_,n1?,n2?);
Fill fib = theta_(-1-n) * sign_(n+1) * fib(-n) + theta_(n-1) * fibimpl(n-2,1,1);
Fill fibimpl = theta_(-n) * n2 + thetap_(n) * fibimpl(n-1,n2,n1+n2);

L F = fib(100);
Print;
.end
```

This computes the 100th Fibonacci number:

F = 354224848179261915075;

### 2.7 Expression database using ArgToExtraSymbol

FORM has tables that map integers (keys) to expressions (values). An arbitrary expression cannot be a key of a table in a direct way. However, the new command ArgToExtraSymbol with the tonumber option gives a way to map an arbitrary expression into an integer, leading to the use of an expression as a key to a table.

Below we give an example where we effectively build a table storing

\[ g(1) \times g(2) \mapsto 100, \]
\[ g(1) \times g(2) \times g(3) \mapsto 200, \]

by using the following code:

```FORM
S n;
CF f,g;
L F = f(g(1)*g(2))*100 + f(g(1)*g(2)*g(3))*200;

argtoextrasymbol tonumber,f;

B f;
Print;
.sort
CTable sparse,values(1);
FillExpression values = F(f);  * store values in table
```
Drop F;
.sort

L G = f(g(1)*g(2)) + f(g(1)*g(2)*g(3)) + f(g(1)*g(2)*g(3)*g(4));

argtoextrasymbol tonumber,f;
id f(n?) = f(extrasymbol_(n))*values(n); * read values in table

Print +s;
.end

The first Print statement gives:

F =
  + f(1) * ( 100 )
  + f(2) * ( 200 );

and the final output is

G =
  + 100*f(g(1)*g(2))
  + 200*f(g(1)*g(2)*g(3))
  + f(g(1)*g(2)*g(3)*g(4))*values(3)


2.8 Partition function

We have added several new convenient statements and functions to FORM 4.2, such as Transform dedup/addargs/mulargs, id_, perm_ and the occurs condition in the If statement. Here we showcase the partitions function, which generates all partitions of a list of arguments into n parts. Each part consists of a function name and a size. This function exploits symmetries of the arguments to make sure that no partition is generated twice. Instead, a combinatorial prefactor is computed.

For example, to partition eight elements into three partitions, where the first part is a function \( f_1 \) with three arguments, the second part a function \( f_1 \) with two arguments, and the third part a function \( f_2 \) with three arguments, we write:

S x1,x2,x3;
CF f1,f2;
L F = partitions_(3,f1,3,f1,2,f1,2,f2,3,x1,x1,x1,x1,x2,x2,x2,x3);

Print +s;
.end

F =
  + 18*f1(x1,x1)*f1(x1,x1,x2)*f2(x2,x2,x3)
  + 6*f1(x1,x1)*f1(x1,x1,x3)*f2(x2,x2,x2)
  + 36*f1(x1,x1)*f1(x1,x2,x2)*f2(x1,x2,x3)
  + 36*f1(x1,x1)*f1(x1,x2,x3)*f2(x1,x2,x2)
\[ + 6 f_1(x_1, x_1) f_1(x_2, x_2) f_2(x_1, x_1, x_3) + 18 f_1(x_1, x_1) f_1(x_2, x_2, x_3) f_2(x_1, x_1, x_2) + 12 f_1(x_1, x_1, x_1) f_1(x_1, x_2) f_2(x_2, x_2, x_3) + 4 f_1(x_1, x_1, x_1) f_1(x_1, x_3) f_2(x_2, x_2, x_2) + 12 f_1(x_1, x_1, x_1) f_1(x_2, x_2) f_2(x_1, x_2, x_3) + 12 f_1(x_1, x_1, x_1) f_1(x_2, x_3) f_2(x_1, x_2, x_2) + 72 f_1(x_1, x_1, x_2) f_1(x_1, x_2) f_2(x_1, x_2, x_3) + 18 f_1(x_1, x_1, x_2) f_1(x_1, x_3) f_2(x_1, x_2, x_2) + 36 f_1(x_1, x_1, x_3) f_1(x_1, x_2) f_2(x_1, x_2, x_2) + 18 f_1(x_1, x_1, x_3) f_1(x_2, x_2) f_2(x_1, x_1, x_2) + 36 f_1(x_1, x_1, x_3) f_1(x_2, x_3) f_2(x_1, x_1, x_2) + 36 f_1(x_1, x_2) f_1(x_1, x_2, x_2) f_2(x_1, x_1, x_3) + 12 f_1(x_1, x_2) f_1(x_2, x_2, x_3) f_2(x_1, x_1, x_1) + 12 f_1(x_1, x_2, x_2) f_1(x_2, x_3) f_2(x_1, x_1, x_1) + 4 f_1(x_1, x_3) f_1(x_2, x_2, x_2) f_2(x_1, x_1, x_1) \]

All options for partitions can be found in the manual [5].

3 Summary

We have shown some new features of FORM 4.2. With these features we hope to make the usage of FORM easier and to have improved performance. This version of FORM is the version used by the FORCER [4] package and its supporting software.

Acknowledgements

We would like to thank everyone who has given feedback and reported bugs, in particular Joshua Davies. This work is supported by the ERC Advanced Grant no. 320651, “HEPGAME”.

References

[1] J.A.M. Vermaseren, New features of FORM, [math-ph/0010025](https://arxiv.org/abs/math-ph/0010025).

[2] J. Kuipers, T. Ueda, J.A.M. Vermaseren and J. Villinga, FORM version 4.0, *Comput. Phys. Commun.* **184** (2013) 1453 [arXiv:1203.6543](https://arxiv.org/abs/1203.6543).

[3] FORM 4.2 Changelog, [https://github.com/vermaseren/form/wiki/Release-Notes---Form-4.2.0#changelog---form-420-2017-07-06](https://github.com/vermaseren/form/wiki/Release-Notes---Form-4.2.0#changelog---form-420-2017-07-06).
[4] B. Ruijl, T. Ueda and J.A.M. Vermaseren, Forcer, a FORM program for the parametric reduction of four-loop massless propagator diagrams, arXiv:1704.06650.

[5] FORM Source Repository, https://github.com/vermaseren/form

[6] M. Tentyukov and J.A.M. Vermaseren, The Multithreaded version of FORM, Comput. Phys. Commun. 181 (2010) 1419 [hep-ph/0702279].

[7] FORM Issue Tracker; https://github.com/vermaseren/form/issues

[8] FORM 4.2 Reference Manual, https://github.com/vermaseren/form/releases/download/v4.2.0/form-4.2.0-manual.pdf

[9] F. Herzog, B. Ruijl, T. Ueda, J.A.M. Vermaseren and A. Vogt, FORM, Diagrams and Topologies, PoS LL2016 (2016) 073 [arXiv:1608.01834].

[10] J. Kuipers, T. Ueda and J.A.M. Vermaseren, Code Optimization in FORM, Comput. Phys. Commun. 189 (2015) 1 [arXiv:1310.7007].

[11] B. Ruijl, J. Vermaseren, A. Plaat and H.J. van den Herik, Combining Simulated Annealing and Monte Carlo Tree Search for Expression Simplification, Proceedings of ICAART Conference 2014 1 (2014) 724 [arXiv:1312.0841].

[12] B. Ruijl, J. Vermaseren, A. Plaat and J.v.d. Herik, HEPGAME and the Simplification of Expressions, arXiv:1405.6369.

[13] B. Ruijl, A. Plaat, J. Vermaseren and J. van den Herik, Why Local Search Excels in Expression Simplification, arXiv:1409.5223.